unittest
{
	assert( [ "3", "2 1 4", "2", "1 1", "2 3" ].parse.expand.solve == "6\n9" );
	assert( [ "5", "7 2 3 8 5", "3", "4 2", "1 7", "4 13" ].parse.expand.solve == "19\n25\n30" );
}



import std.algorithm;
import std.conv;
import std.range;
import std.stdio;
import std.typecons;

void main()
{
	stdin.byLineCopy.parse.expand.solve.writeln;
}

auto parse( Range )( Range input )
if( isInputRange!Range && is( ElementType!Range == string ) )
{
	auto n = input.front.to!( long );
	input.popFront;
	auto ts = input.front.split.to!( long[] );
	input.popFront;
	auto m = input.front.to!( long );
	auto pxs = new long[][]( m, 2 );
	foreach( ref px; pxs )
	{
		input.popFront;
		px = input.front.split.to!( long[] );
	}
	
	return tuple( ts, pxs );
}

auto solve( long[] ts, long[][] pxs )
{
	// ドリンクを飲まなかった時の合計必要時間
	auto tt = ts.sum;
	
	// 各ドリンクを飲んだ時の合計必要時間
	auto anss = new long[ pxs.length ];
	foreach( ref ans, px; lockstep( anss, pxs ) )
	{
		ans = tt - ts[ px[ 0 ] - 1 ] + px[ 1 ];		// 合計から飲む前の時間を引いて飲んだ後の時間を足す
	}
	
	return anss.to!( string[] ).join( "\n" );
}
